Lösungen

1

```
Remove["Global`*"]
```

```
M = {{2, 1}, {4, 3}, {1, 2}};
x[x1_, x2_] := {x1, x2};
b = {4, 3, 5};
MatrixForm[Transpose[{M.x[x1, x2]}]] == MatrixForm[Transpose[{b}]]
```

$$\begin{pmatrix} 2\,x1 + x2 \\ 4\,x1 + 3\,x2 \\ x1 + 2\,x2 \end{pmatrix} == \begin{pmatrix} 4 \\ 3 \\ 5 \end{pmatrix}$$

```
M // MatrixForm
```

$$\begin{pmatrix} 2 & 1 \\ 4 & 3 \\ 1 & 2 \end{pmatrix}$$

■ a

Erweitere die Matrix um die Kolonne b. Die neue Matrix HMatrix löst das Gleichungssystem HMatrix..{x1, x2, 1}={0,0,0}.
Betrachte an dieser Stelle das System HMatrix..{x1, x2, x3}={0,0,0}, das die eindeutige Lösung {0,0,0} haben muss, wenn HMatrix regulät ist. Dann ist also für x3=1 ungleich 0 keine Lösung möglich.

```
HMatrix = Flatten[{Transpose[M], {-b}}, 1] // Transpose
```

```
{{2, 1, -4}, {4, 3, -3}, {1, 2, -5}}
```

```
HMatrix // MatrixForm
```

$$\begin{pmatrix} 2 & 1 & -4 \\ 4 & 3 & -3 \\ 1 & 2 & -5 \end{pmatrix}$$

Die erweiterte die Matrix ist regulär

```
Det[HMatrix]
```

```
-21
```

```
Det[HMatrix] == 0
```

```
False
```

```
MatrixForm[HMatrix.{x1, x2, x3}] == MatrixForm[Transpose[{b}]]
```

$$\begin{pmatrix} 2\,x1 + x2 - 4\,x3 \\ 4\,x1 + 3\,x2 - 3\,x3 \\ x1 + 2\,x2 - 5\,x3 \end{pmatrix} == \begin{pmatrix} 4 \\ 3 \\ 5 \end{pmatrix}$$

```
Solve[Inverse[HMatrix].{x1, x2, x3} == {0, 0, 0}, {x1, x2, x3}]
```

```
{{x1 → 0, x2 → 0, x3 → 0}}
```

```
Solve[Inverse[HMatrix].{x1, x2, 1} == {0, 0, 0}, {x1, x2}]
```

{}

Es gibt daher keine Lösung.

▪ b

```
MT = Transpose[M]
```

{{2, 4, 1}, {1, 3, 2}}

```
MT // MatrixForm
```

$$\begin{pmatrix} 2 & 4 & 1 \\ 1 & 3 & 2 \end{pmatrix}$$

```
MT.M // MatrixForm
```

$$\begin{pmatrix} 21 & 16 \\ 16 & 14 \end{pmatrix}$$

```
Det[MT.M]
```

38

MT.M ist regulär, hat also eine Inverse, was eine eindeutige Lösung bedeutet.

```
M.MT // MatrixForm
```

$$\begin{pmatrix} 5 & 11 & 4 \\ 11 & 25 & 10 \\ 4 & 10 & 5 \end{pmatrix}$$

```
Det[M.MT]
```

0

M.MT ist nicht regulär, hat also keine Inverse, was keine eindeutige Lösung bedeutet.

Weiter gilt : $(A.B)^T == B^T.A^T \;\; == > \;\; (M^T.M)^T == M^T.(M^T)^T == M^T.M.$
Damit ist $M^T.M$ symmetrisch.

▪ c

```
MT.M.x[x1, x2] == MT.b
```

{21 x1 + 16 x2, 16 x1 + 14 x2} == {25, 23}

```
solv0 = Solve[MT.M.x[x1, x2] == MT.b, {x1, x2}] // Flatten
```

$\left\{x1 \to -\frac{9}{19},\ x2 \to \frac{83}{38}\right\}$

```
% // N
```

{x1 → -0.473684, x2 → 2.18421}

```
p0 = {x1, x2} /. solv0; Remove[x1, x2]
```

- **d**

```
Take[M.x[x1, x2], 2]
```

{2 x1 + x2, 4 x1 + 3 x2}

```
Take[b, 2]
```

{4, 3}

```
solv1 = Solve[Take[M.x[x1, x2], 2] == Take[b, 2], {x1, x2}] // Flatten
```

$\left\{x1 \to \frac{9}{2},\ x2 \to -5\right\}$

```
% // N
```

{x1 → 4.5, x2 → -5.}

```
p1 = {x1, x2} /. solv1; Remove[x1, x2]
```

```
Take[M.x[x1, x2], -2]
```

{4 x1 + 3 x2, x1 + 2 x2}

```
Take[b, -2]
```

{3, 5}

```
solv2 = Solve[Take[M.x[x1, x2], -2] == Take[b, -2], {x1, x2}] // Flatten
```

$\left\{x1 \to -\frac{9}{5},\ x2 \to \frac{17}{5}\right\}$

```
% // N
```

{x1 → -1.8, x2 → 3.4}

```
p2 = {x1, x2} /. solv2; Remove[x1, x2]
```

```
Drop[M.x[x1, x2], {2}]
```

{2 x1 + x2, x1 + 2 x2}

```
Drop[b, {2}]
```

{4, 5}

```
solv3 = Solve[Drop[M.x[x1, x2], {2}] == Drop[b, {2}], {x1, x2}] // Flatten
```

{x1 → 1, x2 → 2}

```
p3 = {x1, x2} /. solv3; Remove[x1, x2]
```

```
Show[Graphics[{Line[{p1, p2, p3, p1}],
   Line[{p1 + 2 (p2 - p1), p2 + 2 (p1 - p2)}], Line[{p2 + 2 (p3 - p2), p3 + 2 (p2 - p3)}],
   Line[{p1 + 2 (p3 - p1), p3 + 2 (p1 - p3)}], PointSize[0.03`], Point[p0],
   Point[p1], Point[p2], Point[p3]}], AspectRatio → Automatic]
```



```
Show[Graphics[{Line[{p1, p2, p3, p1}], PointSize[0.03`],
   Point[p0], Point[p1], Point[p2], Point[p3]}], AspectRatio → Automatic]
```



2

```
Remove["Global`*"]
```

```
M = {{1, 2, 3, 4, 5}, {3, 2, 1, 5, 4}, -2 {1, 2, 3, 4, 5} + 3 {3, 2, 1, 5, 4}, {1, 1, 2, 1, 2}};
M // MatrixForm
```

$$\begin{pmatrix} 1 & 2 & 3 & 4 & 5 \\ 3 & 2 & 1 & 5 & 4 \\ 7 & 2 & -3 & 7 & 2 \\ 1 & 1 & 2 & 1 & 2 \end{pmatrix}$$

```
x = {x1, x2, x3, x4, x5};
b1 = {4, 3, 5, 2};
b2 = {4, 3, -2 * 4 + 3 * 3, 2}
```

```
{4, 3, 1, 2}
```

2 a

```
Solve[M.x == b1, x]
```

```
{}
```

```
Solve[M.x == b2, x] // Transpose // MatrixForm
```

$$\begin{pmatrix} x1 \to -\frac{1}{4} + \frac{3\,x4}{4} + \frac{3\,x5}{4} \\ x2 \to \frac{7}{4} - \frac{17\,x4}{4} - \frac{13\,x5}{4} \\ x3 \to \frac{1}{4} + \frac{5\,x4}{4} + \frac{x5}{4} \end{pmatrix}$$

2 b, c, d

x4 und x5 bleiben als Parameter drin. Man sieht, dass die Lösungsmannigfaltigkeit zweidimensional ist.

Ordnung = Dim(Urbildraum) = 5,
Dim(Loes) = Dim(Kern) = 2,
Rang(M) = Dim(Image) = 3

3 Falsche Lösung ohne Orthonormierung der Vektoren

```
Remove["Global`*"]
```

```
A = Transpose[{a = {1, 2, 3}, b = {3, 2, 1}, c = Cross[a, b]}]; A // MatrixForm
```

$$\begin{pmatrix} 1 & 3 & -4 \\ 2 & 2 & 8 \\ 3 & 1 & -4 \end{pmatrix}$$

3 a

```
Dα[α_] :=
  {{Cos[α Degree], -Sin[α Degree], 0}, {Sin[α Degree], Cos[α Degree], 0}, {0, 0, 1}};
Dα[30] // MatrixForm
```

$$\begin{pmatrix} \frac{\sqrt{3}}{2} & -\frac{1}{2} & 0 \\ \frac{1}{2} & \frac{\sqrt{3}}{2} & 0 \\ 0 & 0 & 1 \end{pmatrix}$$

```
Inverse[A] // MatrixForm
```

$$\begin{pmatrix} -\frac{1}{6} & \frac{1}{12} & \frac{1}{3} \\ \frac{1}{3} & \frac{1}{12} & -\frac{1}{6} \\ -\frac{1}{24} & \frac{1}{12} & -\frac{1}{24} \end{pmatrix}$$

```
% // N // MatrixForm
```

$$\begin{pmatrix} -0.166667 & 0.0833333 & 0.333333 \\ 0.333333 & 0.0833333 & -0.166667 \\ -0.0416667 & 0.0833333 & -0.0416667 \end{pmatrix}$$

```
M1 = A.Dα[30].Inverse[A]
```

$$\left\{\left\{\frac{1}{6}+\frac{1}{6}\left(-\frac{3}{2}-\frac{\sqrt{3}}{2}\right)+\frac{1}{3}\left(-\frac{1}{2}+\frac{3\sqrt{3}}{2}\right),\right.\right.$$
$$-\frac{1}{3}+\frac{1}{12}\left(\frac{3}{2}+\frac{\sqrt{3}}{2}\right)+\frac{1}{12}\left(-\frac{1}{2}+\frac{3\sqrt{3}}{2}\right),\ \frac{1}{6}+\frac{1}{6}\left(\frac{1}{2}-\frac{3\sqrt{3}}{2}\right)+\frac{1}{3}\left(\frac{3}{2}+\frac{\sqrt{3}}{2}\right)\Big\},$$
$$\Big\{-\frac{1}{3}+\frac{1}{6}\left(-1-\sqrt{3}\right)+\frac{1}{3}\left(-1+\sqrt{3}\right),\ \frac{2}{3}+\frac{1}{12}\left(-1+\sqrt{3}\right)+\frac{1}{12}\left(1+\sqrt{3}\right),$$
$$-\frac{1}{3}+\frac{1}{6}\left(1-\sqrt{3}\right)+\frac{1}{3}\left(1+\sqrt{3}\right)\Big\},\ \Big\{\frac{1}{6}+\frac{1}{6}\left(-\frac{1}{2}-\frac{3\sqrt{3}}{2}\right)+\frac{1}{3}\left(-\frac{3}{2}+\frac{\sqrt{3}}{2}\right),$$
$$\left.\left.-\frac{1}{3}+\frac{1}{12}\left(-\frac{3}{2}+\frac{\sqrt{3}}{2}\right)+\frac{1}{12}\left(\frac{1}{2}+\frac{3\sqrt{3}}{2}\right),\ \frac{1}{6}+\frac{1}{6}\left(\frac{3}{2}-\frac{\sqrt{3}}{2}\right)+\frac{1}{3}\left(\frac{1}{2}+\frac{3\sqrt{3}}{2}\right)\right\}\right\}$$

```
M1 // N // MatrixForm
```

$$\begin{pmatrix} 0.471688 & 0.0386751 & 0.605662 \\ -0.544658 & 0.955342 & 0.455342 \\ -0.561004 & -0.127992 & 1.30502 \end{pmatrix}$$

```
M2 = A.Dα[-30].Inverse[A]
```

$$\left\{\left\{\frac{1}{6}+\frac{1}{6}\left(\frac{3}{2}-\frac{\sqrt{3}}{2}\right)+\frac{1}{3}\left(\frac{1}{2}+\frac{3\sqrt{3}}{2}\right),\ -\frac{1}{3}+\frac{1}{12}\left(-\frac{3}{2}+\frac{\sqrt{3}}{2}\right)+\frac{1}{12}\left(\frac{1}{2}+\frac{3\sqrt{3}}{2}\right),\ \frac{1}{6}+\frac{1}{6}\left(-\frac{1}{2}-\frac{3\sqrt{3}}{2}\right)+\frac{1}{3}\left(-\frac{3}{2}+\frac{\sqrt{3}}{2}\right)\right\},\right.$$
$$\left\{-\frac{1}{3}+\frac{1}{6}\left(1-\sqrt{3}\right)+\frac{1}{3}\left(1+\sqrt{3}\right),\ \frac{2}{3}+\frac{1}{12}\left(-1+\sqrt{3}\right)+\frac{1}{12}\left(1+\sqrt{3}\right),\ -\frac{1}{3}+\frac{1}{6}\left(-1-\sqrt{3}\right)+\frac{1}{3}\left(-1+\sqrt{3}\right)\right\},$$
$$\left.\left\{\frac{1}{6}+\frac{1}{6}\left(\frac{1}{2}-\frac{3\sqrt{3}}{2}\right)+\frac{1}{3}\left(\frac{3}{2}+\frac{\sqrt{3}}{2}\right),\ -\frac{1}{3}+\frac{1}{12}\left(\frac{3}{2}+\frac{\sqrt{3}}{2}\right)+\frac{1}{12}\left(-\frac{1}{2}+\frac{3\sqrt{3}}{2}\right),\ \frac{1}{6}+\frac{1}{6}\left(-\frac{3}{2}-\frac{\sqrt{3}}{2}\right)+\frac{1}{3}\left(-\frac{1}{2}+\frac{3\sqrt{3}}{2}\right)\right\}\right\}$$

```
M2 // N // MatrixForm
```

$$\begin{pmatrix} 1.30502 & -0.127992 & -0.561004 \\ 0.455342 & 0.955342 & -0.544658 \\ 0.605662 & 0.0386751 & 0.471688 \end{pmatrix}$$

■ 3 b

```
M1.{3, 8, 6} // N
```

{5.35844, 8.74081, 5.12318}

```
M2.{3, 8, 6} // N
```

{-0.474894, 5.74081, 4.95652}

---

3 Richtige Lösung

```
Remove["Global`*"]
```

■ 3 a

Stelle zuerst mittels Kreuzprodukten und Norm ein Normalsystem her.

```
a = {1, 2, 3} / Norm[{1, 2, 3}];
b = {3, 2, 1} / Norm[{3, 2, 1}];
c = Cross[a, b] / Norm[Cross[a, b]];
b1 = Cross[c, a] / Norm[Cross[c, a]]; Norm[b1] // N
```

1.

```
A = Transpose[{a, b1, c}]; A // MatrixForm
```

$$\begin{pmatrix} \frac{1}{\sqrt{14}} & \frac{4}{\sqrt{21}} & -\frac{1}{\sqrt{6}} \\ \sqrt{\frac{2}{7}} & \frac{1}{\sqrt{21}} & \sqrt{\frac{2}{3}} \\ \frac{3}{\sqrt{14}} & -\frac{2}{\sqrt{21}} & -\frac{1}{\sqrt{6}} \end{pmatrix}$$

```
A // N // MatrixForm
```

$$\begin{pmatrix} 0.267261 & 0.872872 & -0.408248 \\ 0.534522 & 0.218218 & 0.816497 \\ 0.801784 & -0.436436 & -0.408248 \end{pmatrix}$$

```
Dα[α_] :=
  {{Cos[α Degree], -Sin[α Degree], 0}, {Sin[α Degree], Cos[α Degree], 0}, {0, 0, 1}};
Dα[30] // MatrixForm
```

$$\begin{pmatrix} \frac{\sqrt{3}}{2} & -\frac{1}{2} & 0 \\ \frac{1}{2} & \frac{\sqrt{3}}{2} & 0 \\ 0 & 0 & 1 \end{pmatrix}$$

```
Inverse[A] // MatrixForm
```

$$\begin{pmatrix} \frac{1}{\sqrt{14}} & \sqrt{\frac{2}{7}} & \frac{3}{\sqrt{14}} \\ \frac{4}{\sqrt{21}} & \frac{1}{\sqrt{21}} & -\frac{2}{\sqrt{21}} \\ -\frac{1}{\sqrt{6}} & \sqrt{\frac{2}{3}} & -\frac{1}{\sqrt{6}} \end{pmatrix}$$

```
% // N // MatrixForm
```

$$\begin{pmatrix} 0.267261 & 0.534522 & 0.801784 \\ 0.872872 & 0.218218 & -0.436436 \\ -0.408248 & 0.816497 & -0.408248 \end{pmatrix}$$

```
M1 = A.Dα[30].Inverse[A] // N
```

```
{{0.888355, 0.159466, 0.430577},
 {-0.248782, 0.955342, 0.159466}, {-0.385919, -0.248782, 0.888355}}
```

```
M1 // N // MatrixForm
```

$$\begin{pmatrix} 0.888355 & 0.159466 & 0.430577 \\ -0.248782 & 0.955342 & 0.159466 \\ -0.385919 & -0.248782 & 0.888355 \end{pmatrix}$$

```
M2 = A.Dα[-30].Inverse[A] // N
```

```
{{0.888355, -0.248782, -0.385919},
 {0.159466, 0.955342, -0.248782}, {0.430577, 0.159466, 0.888355}}
```

```
M2 // N // MatrixForm
```

$$\begin{pmatrix} 0.888355 & -0.248782 & -0.385919 \\ 0.159466 & 0.955342 & -0.248782 \\ 0.430577 & 0.159466 & 0.888355 \end{pmatrix}$$

3 b

```
M1.{3, 8, 6} // N
```

```
{6.52426, 7.85318, 2.18211}
```

```
M2.{3, 8, 6} // N
```

```
{-1.64071, 6.62844, 7.89759}
```

■ 3 c Drehung des Punktes P = {3, 8, 6} um die Achse c

```
OP = {3, 8, 6};
Mα[α_] := A.Dα[α].Inverse[A];
OPα[α_] := Mα[α].OP;
par1 = ParametricPlot3D[{OPα[α], α / 18 c }, {α, -180, 180}]
```



4

```
Remove["Global`*"]
```

■ 4 a

```
v1 = {-1, 3}; v2 = {2, 1}; MV = Transpose[{v1, v2}];
De = {{1, 0}, {0, -2}};
p1 = {1, 1}; p2 = {2, 1};
p3 = {2, 2}; p4 = {1, 2};
M = MV.De.Inverse[MV];
M // MatrixForm
```

$$\begin{pmatrix} -\frac{11}{7} & -\frac{6}{7} \\ -\frac{9}{7} & \frac{4}{7} \end{pmatrix}$$

```
M // N // MatrixForm
```

$$\begin{pmatrix} -1.57143 & -0.857143 \\ -1.28571 & 0.571429 \end{pmatrix}$$

4 b

```
q1 = M.p1; q2 = M.p2; q3 = M.p3; q4 = M.p4;
{q1, q2, q3, q4} // Transpose // MatrixForm
```

$$\begin{pmatrix} -\frac{17}{7} & -4 & -\frac{34}{7} & -\frac{23}{7} \\ -\frac{5}{7} & -2 & -\frac{10}{7} & -\frac{1}{7} \end{pmatrix}$$

```
% // N // MatrixForm
```

$$\begin{pmatrix} -2.42857 & -4. & -4.85714 & -3.28571 \\ -0.714286 & -2. & -1.42857 & -0.142857 \end{pmatrix}$$

▪ 4 c

```
Show[Graphics[{Line[{p1, p2, p3, p4, p1}], Line[{q1, q2, q3, q4, q1}], Line[{3 v1, -3 v1}],
    Line[{3 v2, -3 v2}], Line[{p1, p1 - 2 (p2 - p1)}], Line[{q1, q1 - 2 (q2 - q1)}],
    Line[{p1, p1 - 4 (p4 - p1)}], Line[{q1, q1 - 4 (q4 - q1)}]}], AspectRatio → Automatic]
```



5 Siehe spezieller Output

Gegeben sind zwei sich berührende Reibräder oder Zahnräder (Kreise K1, K2) mit den Mittelpunkten M1 und M2 und den Radien r1, r2.
M1=(0,0), M2=(10,0). Die Radien sind vernünftig wählbar.
Auf K1 befindet sich ein Punkt P1 und auf K2 ein Punkt P2 (ungleich Mittelpunkte und nicht auf dem Rand).
Ausserhalb der Kreise in genügendem Abstand befindet sich ein Punkt Q.
Von P1 nach Q und von P2 nach Q sind Stangen angebracht, welche in P1, P2 und Q miteinander resp. mit den Rädern gelenkig verbinden sind.
Nun beginnt das Rad K1 zu drehen. Wegen der Kraftübertragung (Reibrad oder Zahnrad) dreht auch K2 mit.
Gesucht ist eine graphische Darstellung der Kurve, auf welcher Q bei der Drehung der Räder beschreibt.
Aufgabe: ......

```
Remove["Global`*"]
```

```
M1 = {0, 0}; M2 = {10, 0}; a = 4;
r1 = a; r2 = 10 - a;
P1 = {2, 1}; P2 = {14, -1};
Q = {6, 14};
Show[Graphics[{Circle[M1, r1], Circle[M2, r2], Circle[M1, 0.5`], Circle[M2, 0.5`],
   PointSize[0.03`], Point[M1], Point[M2], Point[P1], Point[P2], PointSize[0.05`],
   Point[Q], Line[{M1, M2}], Line[{P1, Q, P2}]}], AspectRatio → Automatic]
```



```
M1 = {0, 0}; M2 = {10, 0}; a = 4;
r1 = a; r2 = 10 - a;
P1 = {2, 1}; P2 = {12, -2};
H1 = M1 + r1 {Cos[π + 1], Sin[π + 1]};
H2 = M2 + r1 {Cos[π + 1], Sin[π + 1]};
Q = {6, 10};
Show[Graphics[
  {Circle[M1, r1], Circle[M2, r1], Circle[M1, 0.5`], Circle[M2, 0.5`], PointSize[0.03`],
   Point[M1], Point[M2], Point[P1], Point[P2], Point[H1], Point[H2], PointSize[0.05`],
   Point[Q], Line[{M1, M2}], Line[{P1, Q, P2}], Line[{H1, H2}]}], AspectRatio → Automatic]
```



```
M1 = {0, 0}; M2 = {10, 0}; a = 4;
r1 = a; r2 = 10 - a;
P1 = {2, 1}; P2 = {9, 8};
Q = {6, 9};
Show[Graphics[{Circle[M1, r1], Circle[M1, 0.5`], Circle[M2, 0.5`],
   PointSize[0.03`], Point[M1], Point[M2], Point[P1], Point[P2],
   PointSize[0.05`], Point[Q], Line[{P1, Q, P2, M2}]}], AspectRatio → Automatic]
```



http://zirkel-und-lineal.softonic.de/

http://www.wintotal.de/Software/index.php?id=2716

http://www.chip.de/downloads/Zirkel-und-Lineal_18149306.html

http://mathsrv.ku-eichstaett.de/MGF/homes/grothmann/zirkel/

Dynamische Geometrie (Versionen deutsch oder Englisch. Laufen mit Java direkt im Internet-Browser)

Einführung (Video-Film mit Sprache) aufrufbar auf unter Videos, Einführung auf http://zirkel-und-lineal.softonic.de/ oder direkt http://mathsrv.ku-eichstaett.de/MGF/homes/grothmann/zirkel/doc_de/index.html)

Das Programm läuft direkt per Web-Start (mit Java) oder per Download und Installation.